ME 201/MTH 281/ME400/CHE400 Far Field for the Laplace Equation *Mathematica* 7

1. Introduction

In this notebook we consider the far-field behavior of the solution of the boundary value problem given below for the Laplace equation in a two-dimensional semi-infinite region. By far-field behavior, we mean the behavior of the solution as we move far from the boundary on which the inhomogenous boundary condition is specified.

$$\frac{\partial^2 \Phi}{\partial x^2} + \frac{\partial^2 \Phi}{\partial y^2} = 0 \ , \ 0 < x < L, \ y > 0 \ ,$$
$$\text{with } \Phi(0, y) = 0 \ , \ \Phi(L, y) = 0 \ , \ \Phi(x, 0) = f(x), \text{ and } \Phi \underset{y\to\infty}{\longrightarrow} 0 \ . \qquad (1)$$

We obtained the solution to this problem in class by separation of variables. The result is

$$\Phi(x,y) = \sum_{n=1}^{\infty} C_n e^{-n\pi y/L} \sin(n\pi x/L) \ , \text{ where } C_n = \frac{2}{L}\int_0^L f(x) \sin(n\pi x/L)\, dx \ . \qquad (2)$$

For interpretative purposes, it is helpful to compare the series for the solution with the series for the boundary function *f*:

$$f(x) = \sum_{n=1}^{\infty} C_n \sin(n\pi x/L) \ . \qquad (3)$$

We may describe the solution (2) in the following terms. The boundary function $f(x)$ is resolved into Fourier modes $\sin(n\pi x/L)$ with amplitude C_n. The solution for Φ is constructed by letting each mode decay with height by the factor $e^{-n\pi y/L}$, and then reassembling those decaying modes into the series (2). We may abstract a simple scaling rule from this result. We observe that for the mode $\sin(n\pi x/L)$, the half wavelength -- which is a reasonable measure of the scale of variation of this function -- is L/n. From the y-dependent part of the solution for Φ, we see that this mode decays with height with an e-folding scale of $L/n\pi$, which is $\frac{1}{\pi}$times the half-wavelength. Thus high-frequency wiggles on the boundary decay with height much more rapidly than low frequency wiggles. This is the basis for the far-field approximation discussed below. A qualitative summary of this result is as follows: Boundary variations on a scale l penetrate into the interior a distance of the order of l.

In the remainder of this notebook, we illustrate these ideas with graph sequences. In section 2, we consider a far-field approximation to the solution, and show that different boundary conditions can lead to the same far-field approximation. In section 3, we illustrate the limited penetration into the interior of high-frequency boundary components.

2. Far-Field Approximation

If we examine the field Φ some distance above the boundary $y = 0$, the higher harmonics will have essentially disappeared. Specifically, if we are at a height $y = L$, then the first harmonic has decayed by $e^{-\pi}$, the second harmonic by $e^{-2\pi}$, the third harmonic by $e^{-3\pi}$, etc. In the range $y \geq L$, it is reasonable to obtain a simple approximation to the solution by retaining only the first term in the series, and this is our far-field approximation, which we denote by Φ_f. It is given by

$$\Phi_f = C_1\, e^{-\pi y/L} \sin(\pi x/L) \;. \qquad (4)$$

Of course if C_1happens to be zero, then we would have to go to the next non-zero term of the series to get a useful approximation.

Far-field approximations are related to an interesting type of problem called an inverse problem. In general terms, an inverse problem is a situation in which we attempt to infer the sources of a field from measurements of the field outside the source region. In the present problem, the sources of the field are on the boundary $y = 0$ and $0 \leq x \leq L$ -- that is, we may think of the source in this problem as the specified function $f(x)$. Knowledge of f is tantamount to knowledge of $C_n$for all n. From our discussion leading to the far-field approximation (4), we see that as we move further from the boundary, we are less likely to be able to reconstruct f from our measurements. How well we do depends also on how accurately we can measure Φ.

It is clear from (4) that if we are sufficiently far from the boundary, two different sources f could look alike, if they have the same value of C_1. We construct an example illustrating this, in which we look simultaneously at three different solutions, corresponding to three different boundary conditions. In these calculations, we set $L = 1$. The three boundary condtions are f1[x] equal to 1, f2[x] equal to 2 on the left half interval and 0 on the right half interval, and f3[x] equal to $(4/\pi)\sin[\pi x/L]$.

```
L = 1.0;

f1[x_] := 1

f2[x_] := If[(x<=0.5),(2),(0)]

f3[x_] := (4/π)*Sin[π*x/L]
```

The Fourier sine coefficients for these functions are easily calculated. We have

```
C1[n_] := If[OddQ[n], 4/(n*π), 0]

C2[n_] := (4/(n*π)) * (1 - Cos[(n*π)/2])

C3[n_] := If[n == 1, 4/(n*π), 0]
```

These boundary functions have been chosen so that they have the same C_1:

```
C1[1]
```

$\frac{4}{\pi}$

```
C2[1]
```

$\frac{4}{\pi}$

```
C3[1]
```

$\frac{4}{\pi}$

Now we construct the solutions corresponding to each initial condition. For each solution, we include an argument *k* which is the number of terms to keep in the partial sum. It is worth noting that the exact solution Φ3 is also the far-field approximation for Φ1 and Φ2.

```
Φ1[x_,y_,k_] := Sum[C1[n]*Exp[-n*π*y/L]*
            Sin[n*π*x/L],{n,1,k}]

Φ2[x_,y_,k_] := Sum[C2[n]*Exp[-n*π*y/L]*
            Sin[n*π*x/L],{n,1,k}]

Φ3[x_,y_] := C3[1]*Exp[-π*y/L]*
            Sin[π*x/L]
```

Now we define the solution in a computationally convenient way. For $y = 0$, we use the given boundary functions. For $y < 0.1L$ we use 100 terms in the series, and for $y \geq 0.1L$ we use 10 terms in the series. A more sophisticated setup would actually calculate the number of terms needed for any given *y*.

```
sol1[x_,y_] := If[(y == 0),(f1[x]),
    (If[(y < 0.1*L),(Φ1[x,y,100]),(Φ1[x,y,10])])]

sol2[x_,y_] := If[(y == 0),(f2[x]),
    (If[(y < 0.1*L),(Φ2[x,y,100]),(Φ2[x,y,10])])]

sol3[x_,y_] := If[(y == 0),(f3[x]),(Φ3[x,y])]
```

Now we define a function field[y] which produces a graph of the three solutions at the height *y*. The first solution is plotted in red, the second in green and the third in blue.

```
field[y_] := Plot[{sol1[x,y],sol2[x,y],sol3[x,y]},
            {x,0,L},AxesLabel->{"x","Φ"},ImageSize->250,
            PlotLabel->Row[{"y = ",PaddedForm[y,{5,2}]}],
            PlotStyle->{{RGBColor[1,0,0],Thickness[0.004]},{RGBColor[0,1,0],Thi
            RGBColor[0,0,1],Thickness[0.004]}},PlotRange->{0,2.01}]
```

Now we use a Do loop to construct a graph sequence in which *y* is increased gradually. We let *y* run from 0 to *L* in increments of 0.01*L, thus 101 graphs. The printed version of this notebook shows only the initial graph in this sequence.

```
Do[Print[field[i*0.01]],{i,0,100}]
```



If you animate the graph sequence, you will see what happens to the three solutions as we move upward away from the boundary. They eventually coalesce and then decay to zero. The amplitudes are very small for $y = L$, the last graph in the series.

For visualization in the printed version of this notebook, we construct an abbreviated sequence of 6 graphs, with y running from 0 to 1 in increments of 0.2.

```
Do[Print[field[i*0.2]],{i,0,5}]
```




y = 0.20
Φ
2.0
1.5
1.0
0.5
0.0
0.2
0.4
0.6
0.8
1.0
x


y = 0.40
Φ
2.0
1.5
1.0
0.5
0.0
0.2
0.4
0.6
0.8
1.0
x


y = 0.60
Φ
2.0
1.5
1.0
0.5
0.0
0.2
0.4
0.6
0.8
1.0
x





We see from either graph sequence that the green solution (Φ2) approaches the far-field limit more slowly than the red solution (Φ1). We can see why if we examine the Fourier coefficients for the two solutions.

```
C1[2]
```

0

```
C1[3]
```

$$\frac{4}{3\pi}$$

```
C2[2]
```

$$\frac{4}{\pi}$$

```
C2[3]
```

$$\frac{4}{3\pi}$$

Thus in the solution Φ1, the second harmonic is absent, so the approach to the far field is determined by the 3rd harmonic. For Φ2, the second harmonic is present, and gives a slower approach to the far-field approximation simply because the second harmonic decays more slowly than the third harmonic.

■ 3. Rapid Decay of High-Frequency Boundary Components

We consider here a graph sequence which shows the rapid decay of high boundary harmonics as we move away from the boundary. We do this by considering two different solutions. One solution corresponds to a boundary function equal to just the first harmonic. In the second solution we add to this a very high-frequency harmonic -- in this case the $n = 20$ harmonic. (We could think of this as representing high-frequency noise in the boundary function.) We take the amplitude of the high frequency harmonic to be 10% of the amplitude of the fundamental. We construct a movie showing the variation of the solution as we move upward away from the boundary. As the movie shows, the part of the solution associated with the high harmonic disappears very rapidly with height.

We now define the two solutions, which we call Φbase (base solution) and Φn (solution with noise).

```
Φbase[x_,y_] := Exp[-π*y/L]*Sin[π*x/L]

Φn[x_,y_] := Φbase[x,y] + amp*Exp[-k*π*y]*Sin[k*π*x]
```

We continue to take $L = 1$. We set the noise mode number at 20, and the amplitude at 0.1.

```
L = 1.0;

amp = 0.1;

k = 20.0;
```

We define a function graphn[y] which produces a plot of the two solutions at the height y. We plot the base solution in red, and the solution with noise in green. In the printed version of the notebook, only the first graph of the sequence is shown.

```
graphn[y_] := Plot[{Φn[x,y],Φbase[x,y]},
            {x,0,L},AxesLabel->{"x","Φ"},ImageSize->250,
            PlotLabel->Row[{"y =",PaddedForm[y,{5,3}]}],
            PlotStyle->{{RGBColor[0,1,0],Thickness[0.004]},{RGBColor[1,0,0],Thi
            PlotRange->{-amp,1.+ amp}]

Do[Print[graphn[n*0.001]],{n,0,100}]
```



The results seen by animating the graph sequence are consistent with the e-folding penetration scales. For the basic solution with only the first harmonic, the e-folding scale is

L / π

0.31831

For the noise, which is the 20th harmonic, the e-folding scale is

L / (20 * π)

0.0159155

As the graph sequence shows, the noise is for the most part gone by the time the height has reached two or three times this last e-folding scale -- that is, at about $y = 0.05$. For the printed version of the notebook, we show an abbreviated sequence of 6 graphs with y running from 0 to 0.05 in increments of 0.01.

```
Do[Print[graphn[n*0.01]],{n,0,5}]
```




y = 0.010
Φ
1.0
0.8
0.6
0.4
0.2
0.2
0.4
0.6
0.8
1.0
x


y = 0.020
Φ
1.0
0.8
0.6
0.4
0.2
0.2
0.4
0.6
0.8
1.0
x


y = 0.030
Φ
1.0
0.8
0.6
0.4
0.2
0.2
0.4
0.6
0.8
1.0
x

y = 0.040



y = 0.050